# NAG Fortran Library Routine Document

# F07JGF (DPTCON)

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

## 1 Purpose

F07JGF (DPTCON) computes the reciprocal condition number of a real $n$ by $n$ symmetric positive-definite tridiagonal matrix $A$, using the $LDL^{\mathrm{T}}$ factorization returned by F07JDF (DPTTRF).

## 2 Specification

```
SUBROUTINE F07JGF (N, D, E, ANORM, RCOND, WORK, INFO)
INTEGER             N, INFO
double precision    D(*), E(*), ANORM, RCOND, WORK(*)
```

The routine may be called by its LAPACK name ***dptcon***.

## 3 Description

F07JGF (DPTCON) should be preceded by a call to F07JDF (DPTTRF), which computes a modified Cholesky factorization of the matrix $A$ as

$$A = LDL^{\mathrm{T}},$$

where $L$ is a unit lower bidiagonal matrix and $D$ is a diagonal matrix, with positive diagonal elements. F07JGF (DPTCON) then utilizes the factorization to compute $\left\|A^{-1}\right\|_1$ by a direct method, from which the reciprocal of the condition number of $A$, $1/\kappa(A)$ is computed as

$$1/\kappa_1(A) = 1/\left(\|A\|_1\left\|A^{-1}\right\|_1\right).$$

$1/\kappa(A)$ is returned, rather than $\kappa(A)$, since when $A$ is singular $\kappa(A)$ is infinite.

## 4 References

Higham N J (2002) *Accuracy and Stability of Numerical Algorithms* (2nd Edition) SIAM, Philadelphia

## 5 Parameters

1: N – INTEGER *Input*

*On entry*: $n$, the order of the matrix $A$.

*Constraint*: $\mathrm{N} \geq 0$.

2: D($*$) – ***double precision*** array *Input*

**Note**: the dimension of the array D must be at least $\max(1, \mathrm{N})$.

*On entry*: must contain the $n$ diagonal elements of the diagonal matrix $D$ from the $LDL^{\mathrm{T}}$ factorization of $A$.

3: E(∗) – ***double precision*** array *Input*

**Note**: the dimension of the array E must be at least $\max(1, \mathrm{N} - 1)$.

*On entry*: must contain the $(n-1)$ subdiagonal elements of the unit lower bidiagonal matrix $L$. E can also be regarded as the superdiagonal of the unit upper bidiagonal matrix $U$ from the $U^{\mathrm{T}}DU$ factorization of $A$.

4: ANORM – ***double precision*** *Input*

*On entry*: $\|A\|_1$. ANORM may be computed by calling F06RPF with the argument NORM = 'O', and must be computed either before calling F07JDF (DPTTRF), or else from a copy of the original matrix $A$.

5: RCOND – ***double precision*** *Output*

*On exit*: the reciprocal condition number, $1/\kappa_1(A) = 1/\left(\|A\|_1 \left\|A^{-1}\right\|_1\right)$.

6: WORK(∗) – ***double precision*** array *Workspace*

**Note**: the dimension of the array WORK must be at least $\max(1, \mathrm{N})$.

7: INFO – INTEGER *Output*

*On exit*: INFO = 0 unless the routine detects an error (see Section 6).

## 6 Error Indicators and Warnings

Errors or warnings detected by the routine:

INFO < 0

If INFO = $-i$, the $i$th argument had an illegal value. An explanatory message is output, and execution of the program is terminated.

## 7 Accuracy

The computed condition number will be the exact condition number for a closely neighbouring matrix.

## 8 Further Comments

The condition number estimation requires $O(n)$ floating-point operations.

See Section 15.6 of Higham (2002) for further details on computing the condition number of tridiagonal matrices.

The complex analogue of this routine is F07JUF (ZPTCON).

## 9 Example

To compute the condition number of the symmetric positive-definite tridiagonal matrix $A$ given by

$$A = \begin{pmatrix} 4.0 & -2.0 & 0 & 0 & 0 \\ -2.0 & 10.0 & -6.0 & 0 & 0 \\ 0 & -6.0 & 29.0 & 15.0 & 0 \\ 0 & 0 & 15.0 & 25.0 & 8.0 \\ 0 & 0 & 0 & 8.0 & 5.0 \end{pmatrix}.$$

## 9.1 Program Text

```
*     F07JGF Example Program Text
*     Mark 21 Release. NAG Copyright 2004.
*     .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          NMAX
      PARAMETER        (NMAX=50)
*     .. Local Scalars ..
      DOUBLE PRECISION ANORM, RCOND
      INTEGER          I, INFO, N
*     .. Local Arrays ..
      DOUBLE PRECISION D(NMAX), E(NMAX-1), WORK(NMAX)
*     .. External Functions ..
      DOUBLE PRECISION F06RPF, X02AJF
      EXTERNAL         F06RPF, X02AJF
*     .. External Subroutines ..
      EXTERNAL         DPTCON, DPTTRF
*     .. Executable Statements ..
      WRITE (NOUT,*) 'F07JGF Example Program Results'
      WRITE (NOUT,*)
*     Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) N
      IF (N.LE.NMAX) THEN
*
*        Read the lower bidiagonal part of the tridiagonal matrix A from
*        data file
*
         READ (NIN,*) (D(I),I=1,N)
         READ (NIN,*) (E(I),I=1,N-1)
*
*        Compute the 1-norm of A
*
         ANORM = F06RPF('1-norm',N,D,E)
*
*        Factorize the tridiagonal matrix A
*
         CALL DPTTRF(N,D,E,INFO)
*
         IF (INFO.EQ.0) THEN
*
*           Estimate the condition number of A
*
            CALL DPTCON(N,D,E,ANORM,RCOND,WORK,INFO)
*
*           Print the estimated condition number
*
            IF (RCOND.GE.X02AJF()) THEN
               WRITE (NOUT,99999) 'Estimate of condition number = ',
     +           1.0D0/RCOND
            ELSE
               WRITE (NOUT,99999)
     +           'A is singular to working precision. RCOND = ', RCOND
            END IF
*
         ELSE
            WRITE (NOUT,99998) 'The leading minor of order ', INFO,
     +        ' is not positive definite'
         END IF
       ELSE
         WRITE (NOUT,*) 'NMAX too small'
       END IF
       STOP
*
99999 FORMAT (1X,A,1P,E10.2)
99998 FORMAT (1X,A,I3,A)
      END
```

## 9.2 Program Data

```
F07JGF Example Program Data
  5                           :Value of N
  4.0  10.0  29.0  25.0   5.0 :End of diagonal D
 -2.0  -6.0  15.0   8.0       :End of sub-diagonal E
```

## 9.3 Program Results

```
 F07JGF Example Program Results

 Estimate of condition number =   1.05E+02
```